unittest
{
	assert( [ "4", "0 1 1 2" ].parse.expand.solve == 2 );
	assert( [ "4", "1 1 1 1" ].parse.expand.solve == 0 );
	assert( [ "7", "0 3 2 1 2 2 1" ].parse.expand.solve == 24 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	input.popFront;
	auto ds = input.front.split.to!( long[] );
	return tuple( ds );
}

auto solve( long[] ds )
{
	// 根から順番に構築する
	auto ans = 1L;
	auto p = 1L;
	auto d = 0L;
	foreach( g; ds.sort().group )
	{
		// 子を付けられる適正距離の親が無い
		if( d != g[ 0 ] ) return 0;
		
		// 親子の組み合わせのパターン数を計算
		ans = ans.mulMod( p.powMod( g[ 1 ] ) );
		
		p = g[ 1 ];
		d++;
	}
	return ans;
}

// 掛け算
auto mulMod( long MOD = 998244353L )( long x, long y )
{
	return x * y % MOD;
}

// べき乗
auto powMod( long MOD = 998244353L )( long x, long y )
{
	auto a = 1L;
	for( ; 0 < y; y >>= 1 )
	{
		if( y & 1 ) a = a * x % MOD;
		x = x ^^ 2 % MOD;
	}
	return a;
}
